Stochastic Process (I)

Xiangyi Meng

Accumulated Gaussian random variables

```
ClearAll["Global`*"]

x = Table[NormalDistribution[μ, σ], 5];
TransformedDistribution[x1 + x2 + x3 + x4 + x5,
  {x1 ≈ x[[1]], x2 ≈ x[[2]], x3 ≈ x[[3]], x4 ≈ x[[4]], x5 ≈ x[[5]]}, Assumptions → σ > 0]
```

NormalDistribution[5μ, $\sqrt{5}\,\sigma$]

Wiener process

Wiener Process is also known as Brownian motion, a continuous-time random walk, or integrated white Gaussian noise.

```
ClearAll["Global`*"]
SliceDistribution[WienerProcess[μ, σ], t]
```

```
NormalDistribution[t μ, √t σ]
```

```
itoProcessX = ItoProcess[ⅆX[t] == μ ⅆt + σ ⅆW[t], X[t], {X, 0}, t, W ≈ WienerProcess[0, 1]];
```

Differential form: $dX = \mu dt + \sigma dW$,
(dW is WienerProcess[0,1]; $dW \sim \sqrt{dt}$)

Ito's lemma

What's the differential form of Y=f(X), a function of stochastic process X?

Ito's Lemma: $df(X)=(\frac{\partial f}{\partial t}+\mu\frac{\partial f}{\partial x}+\frac{\sigma^2}{2}\frac{\partial^2 f}{\partial x^2})dt+\sigma\frac{\partial f}{\partial x}dW.$

e.g., supposing Y=Exp[X], we have

$$dY=(\mu+\frac{\sigma^2}{2})Y\,dt+\sigma Y\,dW,$$

which is called geometric Brownian motion.

```
itoProcessY =
  ItoProcess[ⅆ(Y[t]) == (μ + σ^2/2) Y[t] ⅆt + σ Y[t] ⅆW[t], Y[t], {Y, 0}, t, W ≈ WienerProcess[0, 1]];
TransformedProcess[Log[Y[t]], Y ≈ GeometricBrownianMotionProcess[μ + σ^2/2, σ, 1], t];
PDF[%[t], x] // FullSimplify
WienerProcess[μ, σ];
PDF[%[t], x] // FullSimplify
```

$$\frac{e^{-\frac{(x-t\mu)^2}{2t\sigma^2}}}{\sqrt{2\pi}\sqrt{t}\,\sigma}$$

$$\frac{e^{-\frac{(x-t\mu)^2}{2t\sigma^2}}}{\sqrt{2\pi}\sqrt{t}\,\sigma}$$

1) Suppose your portfolio first increases 10%, then decreases 10%. How much will you gain/lose?

1) Suppose your portfolio first decreases 10%, then increases 10%. How much will you gain/lose?

Scaling of PDF

```
sliceDist = NormalDistribution[μ t, σ Sqrt[t]];
PDF[sliceDist, x]
```

$$\frac{e^{-\frac{(x-t\mu)^2}{2t\sigma^2}}}{\sqrt{2\pi}\sqrt{t}\sigma}$$

```
Mean@sliceDist
StandardDeviation@sliceDist
(*Max probablility*)PDF[sliceDist, x] /. x → μ t
```

$t\mu$

$\sqrt{t}\sigma$

$$\frac{1}{\sqrt{2\pi}\sqrt{t}\sigma}$$

However, empirical analysis shows Max[P] ~ $t^{-0.53}$ for daily prices and Max[P] ~ $t^{-0.71}$ for high-frequency prices.

Levy flight

What if the PDF is non-Gaussian?

Independent random variables of that PDF must be additive, e.g., Gaussian, Cauchy...

```
ClearAll["Global`*"];
levyPDF[x_, t_, k_, γ_] := 1/π NIntegrate[Exp[- k q^(-1/γ) t] Cos[q x], {q, 0.0, ∞}]
returns = Differences@Log@FinancialData["SP500", {{2006}, {2010}}, "Value"];
dist = HistogramDistribution[returns - Mean@returns, {0.002}];
Plot[PDF[dist, x], {x, -0.05, 0.05}]
Z = 100.;
fitData = Table[{Z x, Z^-1 PDF[dist, x]}, {x, -0.05, 0.05, 0.002}];
k = κ /. FindFit[fitData, levyPDF[x, 1., κ, -0.71], {κ}, x]

{
  Table[levyPDF[x, 1.0, k, -0.71], {x, -5.0, 5.0, 0.1}],
  Table[Z^-1 PDF[NormalDistribution[0, StandardDeviation@dist], Z^-1 x], {x, -5.0, 5.0, 0.1}],
  Table[Z^-1 PDF[dist, Z^-1 x], {x, -5.0, 5.0, 0.1}]
 };
ListLogPlot[%, PlotLegends → {"Levy", "Gaussian", "Empirical"}]
```

NIntegrate: DoubleExponentialOscillatory has failed to converge for the integrand 1. $e^{-0.435871\,(0.+q)^{1.40845}}$ Cos[3.4 q] over {0, ∞}. DoubleExponentialOscillatory obtained 0.027112152935888024` and 7.03983287257861`*^-6 for the integral and error estimates.

NIntegrate: DoubleExponentialOscillatory has failed to converge for the integrand 1. $e^{-0.435871\,(0.+q)^{1.40845}}$ Cos[3.4 q] over {0, ∞}. DoubleExponentialOscillatory obtained 0.027112152935889963` and 7.039832571372772`*^-6 for the integral and error estimates.



Conclusion

- **1) Gaussian PDF -> Wiener process.**
- **2) Non-Gaussian PDF -> Levy flight.**
- **3) Empirical PDF is fat-tail.**

Correlation between random variables at different t ?
-> Autocorrelation of time series.

To be Continued...